

... CORRELATION PLOT**PURPOSE**

Generates an auto-, cross-, or partial auto-correlation plot.

DESCRIPTION

A correlation plot is a graphical data analysis technique for determining if correlation exists between:

1. lags for a single time series (an autocorrelation plot);
2. lags for a single time series after removing the linear dependence of intermediate lags (a partial autocorrelation plot);
3. lags for 2 time series (a cross-correlation plot).

The correlation plot consists of:

Vertical axis = correlation coefficient (a value between -1 and 1);
 Horizontal axis = lag (an integer between 1 and $n/4$ where n is the number of observations).

In addition, vertical lines are drawn at zero and at levels indicating statistically significant correlation.

The autocorrelation plot is used in time series modeling. It can be used in conjunction with the LET and FIT commands to generate Box-Jenkins Auto Regressive (AR) models. It is also used to test for randomness (e.g., autocorrelation plots are often generated for the residuals from a least squares fit). The partial autocorrelation plot is used (typically with the autocorrelation plot as well) in the model identification stage when developing Box-Jenkins ARMA models. Cross-correlation is used for the more complex case of analyzing two distinct time series to see if they are related.

SYNTAX 1

AUTOCORRELATION PLOT <y> <SUBSET/EXCEPT/FOR qualification>
 where <y> is a variable containing the time series observations to be analyzed;
 and where the <SUBSET/EXCEPT/FOR qualification> is optional.

SYNTAX 2

CROSS-CORRELATION PLOT <y1> <y2> <SUBSET/EXCEPT/FOR qualification>
 where <y1> is a variable containing the observations for the first time series;
 <y2> is a variable containing the observations for the second time series;
 and where the <SUBSET/EXCEPT/FOR qualification> is optional.

SYNTAX 3

PARTIAL AUTOCORRELATION PLOT <y> <SUBSET/EXCEPT/FOR qualification>
 where <y> is a variable containing the time series observations to be analyzed;
 and where the <SUBSET/EXCEPT/FOR qualification> is optional.

EXAMPLES

```
AUTOCORRELATION PLOT Y
CROSS-CORRELATION PLOT Y1 Y2
```

NOTE 1

The partial autocorrelation of lag k is the autocorrelation between z_t and z_{t+k} with the linear dependence of z_{t+1} thru z_{t+k-1} removed.

NOTE 2

DATAPLOT writes some conclusions derived from the correlation plot to the file DPCONF.TEX (the name may vary depending on the operating system). This file is opened when DATAPLOT is initially started. However, only a few commands actually write any information to this file.

NOTE 3

After generating an autocorrelation or cross-correlation plot, the variable YPLOT contains the numerical values for the correlations. If you need to do additional analysis on the correlations, or you simply wish to print the numeric values, copy this variable to a user variable (e.g., LET CORR = YPLOT). The YPLOT variable is overwritten when the next plot is generated.

NOTE 4

The LINE, CHARACTER, SPIKE, and BAR commands can be used to control the appearance of the correlation plot. For example, some analysts prefer to draw correlation plots as spikes from a zero baseline. Trace 1 is the correlation line, trace 2 is the line at zero, and traces 3 and 4 are the lines indicating statistical significance. This is demonstrated in the sample program below.

NOTE 5

The number of lags is determined automatically. If you want to over-ride the default value, enter one of the following commands:

```
LET LAGS = <value>
LET LAG = <value>
LET NUMLAG = <value>
```

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

SPECTRUM	=	Generates a spectral plot.
PERIODOGRAM	=	Generates a periodogram.
COMPLEX DEMODULATION PLOT	=	Generates a complex demodulation plot.
LAG PLOT	=	Generates a lag plot.
PLOT	=	Generates a data or function plot.
4-PLOT	=	Generates a 4-plot for univariate analysis.
CHARACTERS	=	Sets the types for plot characters.
LINES	=	Sets the types for plot lines.
SPIKES	=	Sets the on/off switches for plot spikes.
SUMMARY	=	Generates a table of summary statistics.
LET	=	Generates sine/cosine transformations (and much more).
FIT	=	Carries out a least squares fit.

REFERENCE

"Time Series Analysis: Forecasting and Control," Box and Jenkins, Holden-Day, 1976.

APPLICATIONS


Time Series Analysis

IMPLEMENTATION DATE

Pre-1987 (the PARTIAL AUTOCORRELATION PLOT command was implemented 93/7)


PROGRAM 1

```
. THIS SAMPLE PROGRAM READS THE FILE LEW.DAT IN THE  
. DATAPLOT REFERENCE DIRECTORY. THESE DATA ARE  
. BEAM DEFLECTION DATA.  
.  
SKIP 25  
READ LEW.DAT DEFLECT  
.  
LINE BLANK SOLID DOT DOT  
SPIKE ON  
SPIKE BASE 0  
XLABEL LAG  
YLABEL AUTOCORRELATION  
TITLE AUTOMATIC  
.  
AUTOCORRELATION PLOT DEFLECT
```



PROGRAM 2

```
. THIS SAMPLE PROGRAM READS THE FILE HAYES1.DAT IN THE  
. DATAPLOT REFERENCE DIRECTORY. THESE DATA ARE  
. FIRE RESEARCH SMOKE OBSCURATION.  
.  
SKIP 25  
READ HAYES1.DAT JUNK Y1 Y2  
.  
LINE BLANK SOLID DOT DOT  
SPIKE ON  
SPIKE BASE 0  
XLABEL LAG  
YLABEL CROSS-CORRELATION  
TITLE AUTOMATIC  
.  
CROSS-CORRELATION PLOT Y1 Y2
```



PROGRAM 3

```
. THIS SAMPLE PROGRAM READS THE FILE LEW.DAT IN THE  
. DATAPLOT REFERENCE DIRECTORY. THESE DATA ARE  
. BEAM DEFLECTION DATA.  
.  
SKIP 25  
READ LEW.DAT DEFLECT  
.  
LINE BLANK SOLID DOT DOT DOT DOT  
SPIKE ON  
SPIKE BASE 0  
XLABEL LAG  
YLABEL PARTIAL AUTOCORRELATION  
.  
PARTIAL AUTOCORRELATION PLOT DEFLECT
```

